```
---------------------------
HHC 2018 Programming Contest
---------------------------

Purpose:    Given a user's birthdate, compute the next biorhythm Extrema Date.
Author:     Kerry Veenstra
Date:       Oh, my gosh, is it Sunday already?


-----
Usage
-----

1. Run the program.
2. See the prompt "MM.DDYYYY".
3. Enter the birthday and press R/S.
4. The next Extrema Date is displayed as E=MM/DD/YYYY
5. Press R/S.
6. The physical biorhythm of the next Extrema Date is displayed as PH=nnn
7. Press R/S.
8. The emotional biorhythm of the next Extrema Date is displayed as EM=nnn
9. Press R/S.
10. The intellectual biorhythm of the next Extrema Date is displayed as IN=nnn
11. (optional) Press R/S and go back to step 2.


-------
Details
-------

Biorhythm values for an individual are computed from these three equations,
where t is the number of days since birth:

    P = round(100 sin(2π t ÷ 23))
    E = round(100 sin(2π t ÷ 28))
    I = round(100 sin(2π t ÷ 33))

P, E, and I occasionally hit 100 and -100.  In fact, they hit these values
exactly twice during each of their cycles.  We can use this observation to
simplify computations.

The tables below demonstrate these situations for each of the three cycles:

     t     P   Comment
    ---  ----  ------------------------------
     5    98
     6   100  One maximum per 0-to-22 cycle
     7    94
    ...
    16   -94
    17  -100  One minimum per 0-to-22 cycle
    18   -98
```

t	E	Comment
6	97	
7	100	One maximum per 0-to-27 cycle
8	97	
...		
20	-97	
21	-100	One minimum per 0-to-27 cycle
22	-97	

t	I	Comment
7	97	
8	100	One maximum per 0-to-32 cycle
9	99	
...		
24	-99	
25	-100	One minimum per 0-to-32 cycle
26	-97	

"Extrema Dates" are those dates in which two or three of the biorhythms are minimums or maximums simultaneously. (Let's call these "collisions".) The pattern of Extrema Dates repeats every 23 × 28 × 33 = 21252 days. This program is simplified by using this observation and mapping any number of days since birth to a value in the range 0 to 21251.

If one examines the pattern of Extrema Dates, one sees that there are eight overlapping regular cycles of Biorhythm collisions. This table defines the patterns of these eight collision cycles:

Bio #1	Bio #2	Start Day	Increment	Start of collisions sequence
P	E	63	322	63, 385, 707, 1029, 1351, ...
P	E	259	322	259, 581, 903, 1225, 1547, ...
P	I	190	759	190, 949, 1708, 2467, 3226, ...
P	I	305	759	305, 1064, 1823, 2582, 3341, ...
P	I	454	759	454, 1213, 1972, 2731, 3490, ...
P	I	569	759	569, 1328, 2087, 2846, 3605, ...
E	I	91	462	91, 553, 1015, 1477, 1939, ...
E	I	371	462	371, 833, 1295, 1757, 2219, ...

One can take a days-since-birth value (mapped into the 21252-day cycle) and examine all eight of these collisions cycles for the next Extrema Date.

NOTE: The days-since-birth value of 21189 is the last Extrema Date in the

range 0 to 21251. Any value of days-since-birth ≥ 21189 should report the first Extrema Date of the next cycle.

Program

Prompt for a birthday.

```
01 LBL "BIO"
02 MDY
03 "MM.DDYYYY"
04 PROMPT
```

Compute the number of days since birth. NOTE: the program assumes that the calculator's date is set correctly. Also, if the birthday is not entered as a valid date, the program will stop and display "DATA ERROR Y".

Set R09 = Birthday

```
05 STO 09
06 DATE
07 DDAYS
```

Split the number of days since birth into two values and store:

Set R10 = The number of days that the subject has lived through full 21252-day Extrema-Date cycles.

Set R11 = The remaining number days lived during the final (partial) Extrema-Date cycle.

```
08 STO Z
09 21252
10 STO Z
11 /
12 INT
13 *
14 STO 10
15 -
16 STO 11
```

Determine the next Extrema Date in each of the eight collision cycles. The result will be in R12. Check for the special case of days-since-birth ≥ 21189, which means the next Extrema Date is at the beginning of the next cycle. In that case, return 21252 + 63 = 21315.

Set R12 = The next Extrema Date in the 21252-day cycle.

Note: X already is days-since-birth.

```
17 21315
18 STO 12
19 126
20 -
21 X<=Y?
22 GTO 00
23 322
24 ENTER
25 63
26 XEQ 10
27 322
28 ENTER
29 259
30 XEQ 10
31 759
32 ENTER
33 190
34 XEQ 10
35 759
36 ENTER
37 305
38 XEQ 10
39 759
40 ENTER
41 454
42 XEQ 10
43 759
44 ENTER
45 569
46 XEQ 10
47 462
48 ENTER
49 91
50 XEQ 10
51 462
52 ENTER
53 371
54 XEQ 10
55 LBL 00
```

R12 is the next Extrema Date in the partial cycle. Add in R10 to get the days-since-birth of the next Extrema Date. Then add to the birthday, and display the resulting date. Also, store the resulting ~~date~~ days-since-birth for computing its biorhythms.

Set R15 = days since birth of next Extrema Date

```
56 RCL 09
```

```
57 RCL 12
58 RCL 10
59 +
60 STO 15
61 DATE+
62 "E="
63 FIX 6
64 ADATE
65 PROMPT
```

Show the biorhythms for that date.

```
66 "PH="
67 23
68 XEQ 11
69 "EM="
70 28
71 XEQ 11
72 "IN="
73 33
74 XEQ 11
```

End of main routine. Restart.

```
75 GTO "BIO"
```

Subroutine 10:

Find the next Extrema Date for the days-since-birth in R11.
Consider only the collision sequence that starts with X and
has an increment of Y. Save the most recent collision
date in R12.

```
Set R13 = start
Set R14 = increment

if R11 < R13 then
    next = R13
else
    next = R13 + R14 * ROUNDDOWN((R11 - R13) / R14 + 1, 0))
```

```
76 LBL 10
77 STO 13
78 RCL 11
79 X<>Y
80 X>Y?
81 GTO 00
```

X = R13, Y = R11, Z = increment

```
82 -
```

X = R11 - R13, Y = increment

```
83 X<>Y
84 STO 14
```

```
85 /
                    X = (R11 - R13) / R14
86 1
87 +
                    X = (R11 - R13) / R14 + 1
88 ENTER
89 FRC
90 -
                    X = ROUNDDOWN(R11 - R13) / R14 + 1, 0)
91 RCL 14
92 *
93 RCL 13
94 +
95 LBL 00
                    Set R12 = min(X, R12)
96 RCL 12
97 X>Y?
98 X<>Y
99 STO 12
100 RTN
```

Subroutine 11:

Compute biorhythms for the days-since-birth in R15 using the cycle length in X. Display using the prefix in ALPHA.

```
101 LBL 11
102 2
103 PI
104 *
105 RCL 15
106 *
107 X<>Y
108 /
109 RAD
110 SIN
111 100
112 *
113 FIX 0
114 RND
115 ARCL X
116 PROMPT
117 RTN
```

unneeded